15-150 Fall 2019

Stephen Brookes

Lecture 6
Sorting lists of integers
1 Outline

In this lecture we put together the themes introduced so far

- recursive function definition
- inductive proof techniques
- specifications, correctness, and efficiency

and we develop functional programs that implement the insertion sort and mergesort algorithms for sorting lists of integers. You should be able to do similarly with other well known sorting algorithms, such as quicksort. Later we will generalize to sorting lists of other data, with respect to different orderings; and we’ll talk about sorting trees and other structured data. We’ll even develop some parallel sorting functions.

For now, it would help to review your knowledge of basic properties of the integers and the “usual” $<$ ordering. If you are not familiar with the notation and basic facts you should consult a discrete math text or look online! You have probably been using many of these properties without much thought, but here we will need to be explicit so that you see their relevance in proving correctness of programs. In particular, for each pair of integers $m$ and $n$ exactly one of the following three relationships holds: either $m < n$, or $n < m$, or $m = n$. We write $x \leq y$ for $x < y$ or $x = y$, and we call $\leq$ the less-than-or-equal-to relation. It satisfies the “transitive” property: $x \leq y$ & $y \leq z$ implies $x \leq z$. It is also anti-symmetric, in that $x \leq y$ & $y \leq x$ implies $x = y$. Note also that for any pair of integers we always have either $x \leq y$ or $y \leq x$; Wikipedia calls this the connex property, but I am unfamiliar with that terminology! I prefer to call it “totality” of the $\leq$ relation. In summary, this combination of properties (anti-symmetry, transitivity, and totality) characterizes what is known as a linear (or total) ordering. So what we’re saying is that the usual less-than-or-equal relation on the integers is a total ordering.

It’s often more convenient to refer to $<$ (the strictly less-than relation) instead of $\leq$, but the two notions are intimately connected so there shouldn’t be any confusion. Of course, the $<$ relation on the integers also satisfies transitivity: $x < y$ and $y < z$ implies $x < z$.

And it’s common to use the word “order” instead of “ordering”, and I reserve the right to use these terms interchangeably since I’ve been doing that for years now!
2 Remarks

In the code below we provide clear and accurate specifications but we don’t always use the REQUIRES and ENSURES format. In Lecture 5 we discussed much of this code. These notes supplement what was said and done in class, offering a different perspective and often more explicit detail. Some of the discussion of work in these notes may occur in the next Lecture 6, depending on time.

3 Background

Comparison in ML

We introduce the following type and function; each is already pre-defined in SML, but we give explicit definitions since we use them here for the first time. (Actually the function compare is called Int.compare in SML. We give it a shorter name just for convenience.)

datatype order = LESS | EQUAL | GREATER

(* Comparison for integers *)

(* compare : int * int -> order *)
fun compare(x:int, y:int):order =
    if x<y then LESS else
    if y<x then GREATER else EQUAL

(* compare(x,y)=LESS if x<y *)
(* compare(x,y)=EQUAL if x=y *)
(* compare(x,y)=GREATER if x>y *)

The first line above is a simple form of datatype definition. It introduces a user-defined type named order. Later in the semester we will see more sophisticated datatype definitions. The ability to define your own types and have them fit in seamlessly with the rest of the ML type discipline is a very powerful feature of the programming language.

The type order has just three (syntactic) values, written LESS, EQUAL, and GREATER. You can test expressions of this type for equality (using =) or
use them in case-expressions to do different things based on a comparison between two integer values. (The type order is an equality type.) By the way, the < operator in ML, of type int * int -> bool, behaves just like the “usual” less-than ordering on the integers in math. In math we write \( x \leq y \) for “\( x \) is less-than-or-equal-to \( y \)”, but in ML we use \( x \leq y \). Also int is an equality type, so we can use = to test for equality of integer expressions.

**Linear ordering**

As discussed in the Introduction, the \( \leq \) operator on integers has some well known properties. These are crucial in the design and analysis of our ML implementation of sorting algorithms. In particular, \( \leq \) is a linear ordering. This means that for all integers \( x, y, z \):

- \( x \leq y \) & \( y \leq x \) implies \( x = y \) (antisymmetry)
- \( x \leq y \) & \( y \leq z \) implies \( x \leq z \) (transitivity)
- \( x \leq y \) or \( y \leq x \) (totality)

**Sorted lists**

A list of integers is sorted if each item in the list is \( \leq \) all items that occur later in the list. Here is an ML function that checks for this property.

We only use this function in specifications!

\[
(* \text{sorted} : \text{int list} \to \text{bool} *)
\]

\[
\text{fun sorted} \ [\ ] = \text{true}
\]

\[
\quad | \text{sorted} \ [x] = \text{true}
\]

\[
\quad | \text{sorted} \ (x::y::L) =
\]

\[
\quad \quad (\text{compare}(x,y) <> \text{GREATER}) \ \text{andalso} \ \text{sorted}(y::L)
\]

\[
\]

\[
(* \text{REQUIRES} \ L \text{ is an integer list} *)
\]

\[
(* \text{ENSURES} \ \text{sorted} \ L = \text{true} \ iff \ L \text{ is a sorted list} *)
\]

Note the use of \( <> \), which is ML for “not equal to” and can be used on values of type order. (You can use \( <> \) and = on values of any equality type, e.g. on integers.) The expression \( \text{compare}(x,y) <> \text{GREATER} \) evaluates to true if the value of \( \text{compare}(x, y) \) is LESS or EQUAL, and evaluates to false if \( x > y \).
Examples:

\[
\text{sorted } [1,2,3] = \text{true} \\
\text{sorted } [3,2,1] = \text{false}
\]

Make sure you see the relevance of the linear ordering properties here: they are the reason why this \text{sorted} function behaves as described! And they also explain why it’s not necessary in the third clause of the function to check explicitly that \(x\) is not greater than the elements of \(L\); the term \text{sorted}(y::L) checks that \(y\) is less-than-or-equal to the elements of \(L\), and the knowledge that \(x<y\) is enough (because of transitivity).

From now on we will say that an integer list \(L\) is \text{sorted} if and only if \text{sorted}(L) evaluates to \text{true}. We will assume you remember the basic properties described above, and we will often take advantage of “obvious” properties, such as: an integer list of length 1 is sorted, and an integer list of form \([x,y]\) is sorted if and only if \(x \leq y\). Usually these facts are easy to prove from the definition of sortedness, and they rely on the basic properties of linear orderings.

## 4 Insertion sort

Here is a function that implements the \text{insertion sort} algorithm. This algorithm is often (informally) described as building up its sorted output list by starting with an empty list and successively inserting the items from the input list, at each stage maintaining the correct sorted order among the items inserted so far. Unfortunately this kind of description is too vague to be precise, using fuzzy words like “building up” and “successively”. A better (more precise) description would be:

To \text{insertion sort} a list: if the list is empty, do nothing; otherwise, (recursively) \text{insertion sort} the tail of the list and then \text{insert} the head.

Of course we need a helper function for \text{inserting} an item into a list, and we can see from the above that we will only need to do an insertion of an element into an already sorted list, making sure to maintain sortedness.

Let’s code this algorithm in SML. We won’t need to use ML functions for extracting heads and tails of lists; instead we’ll use pattern matching.
First we define a helper function for inserting an integer into its proper place in a sorted list. To be clear about what this means, we refer to the familiar notion of permutation. An integer list $A$ is a permutation of a list $B$ if $A$ contains the same items as $B$, possibly in a different order, and each integer occurs the same number of times in $A$ as in $B$. For example, $[1,2,3,1]$ is a permutation of $[2,1,3,1]$ and a permutation of $1::(1::2::[3])$, but not a permutation of $[1,2,3]$.

(* ins : int * int list -> int list *)
(* REQUIRES L is a sorted list of integers *)
(* ENSURES ins (x, L) = a sorted permutation of x::L *)

fun ins (x, [ ]) = [x] 
| ins (x, y::L) = case compare(x, y) of 
  GREATER => y::ins(x, L) 
| _ => x::y::L

Examples:
ins (2, [1,3]) = [1,2,3]
ins (2, [3,1]) = [2,3,1]
ins (2, [1,2,3]) = [1,2,2,3]

As the ins function does a three-way comparison but does the same thing in two outcomes it would be just as acceptable stylistically to use an if-then-else with a single < test:

fun ins (x, [ ]) = [x] 
| ins (x, y::L) = if x > y then y::ins(x, L) else x::y::L

Using ins as a helper, we can implement insertion sort as follows:

(* isort : int list -> int list *)
(* REQUIRES L is an integer list *)
(* ENSURES isort(L) = a sorted permutation of L *)

fun isort [ ] = [ ] 
| isort (x::L) = ins (x, isort L)

In the lecture slides and on the blackboard we sketched a proof that ins satisfies its specification, and that isort satisfies its specification. Be sure to study these proofs (and fill in any missing details). They are a very useful exercise in understanding how recursion and induction work.
A variation

Just for interest, here is a slight variation on this theme:

(* isort2 : int list -> int list *)
fun isort2 [ ] = [ ]
| isort2 [x] = [x]
| isort2 (x::L) = ins (x, isort2 L)

We can show easily that this isort2 function is extensionally equivalent to isort as defined above. (So the singleton clause is irrelevant.) Indeed, it is very straightforward to prove, by induction on \( n \), that:

For all \( n \geq 0 \) and all integer values \( x_1, \ldots, x_n \),

\[
\text{isort} [x_1, \ldots, x_n] = \text{isort2} [x_1, \ldots, x_n].
\]

- For \( n = 0 \), we have \( \text{isort} [ ] = [ ] = \text{isort2} [ ] \).
- For \( n = 1 \), we have

\[
\text{isort} [x_1] = \text{ins}(x_1, \text{isort} [ ]) \quad \text{by def of isort}
= \text{ins}(x_1, [ ]) \quad \text{by def of isort}
= [x_1] \quad \text{by def of ins}
= \text{isort2} [x_1] \quad \text{by def of isort2}
\]

- For the inductive step, suppose \( n > 1 \) and the property holds for all lists of length \( n - 1 \), i.e.

\[(\text{IH}): \text{For all integer values } y_1, \ldots, y_{n-1},
\]

\[
\text{isort} [y_1, \ldots, y_{n-1}] = \text{isort2} [y_1, \ldots, y_{n-1}].
\]

Let \( L \) be an integer list of length \( n \), of the form \( [x_1, \ldots, x_n] \). By the function definitions we have

\[
\text{isort} [x_1, \ldots, x_n] = \text{ins}(x_1, \text{isort} [x_2, \ldots, x_n]) \quad \text{by def of isort}
= \text{ins}(x_1, \text{isort2} [x_2, \ldots, x_n]) \quad \text{by IH}
= \text{isort2} [x_1, x_2, \ldots, x_n] \quad \text{by def of isort2}
\]
Permutations

Here are some SML functions to help understand permutations:

```sml
fun mem(x:int, [ ]) = false
| mem(x, y::L) = (x=y) orelse mem(x,L)
(* ENSURES mem(x,L) = true if x occurs in L, false otherwise *)

fun del(x, y::R) = if x=y then R else y::del(x,R)
(* REQUIRES x occurs in L *)
(* ENSURES del(x,L) = a list containing all items in L except for
the first occurrence of x *)

fun perm([ ], [ ]) = true
| perm(_, [ ]) = false
| perm([ ], _) = false
| perm(x::L, R) = mem(x, R) andalso perm(L, del(x,R))
(* ENSURES perm(L,R) = true if L is a permutation of R, false otherwise *)
```

We say that \( L \) is a permutation of \( R \) if the two lists have the same elements (the same number of times, for each one) but possibly in a different relative order. This is the case if and only if \( \text{perm}(L, R) = \text{true} \).

Some pretty obvious facts:

- The only permutation of [ ] is [ ]. Every list is a permutation of itself.
- \( x :: P \) is a permutation of \( L \) if and only if \( x \) is a member of \( L \) and \( P \) is a permutation of \( \text{del}(x, L) \).
- If \( P \) is a permutation of \( L \), then \( x :: P \) is a permutation of \( x :: L \).
- \( (x::A)@(y::B) \) is a permutation of \( x::y::(A@B) \)
- If \( P \) is a permutation of \( L \), then \( \text{length} \ P = \text{length} \ L \).
- If \( P \) is a permutation of \( L \) and \( Q \) is a permutation of \( P \), then \( Q \) is a permutation of \( L \). (Loosely speaking, “A permutation of a permutation is a permutation.”)

We allow use of these facts without proof.
5 Mergesort

To mergesort a list of integers, if it is empty or a singleton do nothing (it’s already sorted); otherwise split the list into two lists of roughly equal length, mergesort these two lists, then merge these two sorted lists.

Obviously we need helper functions for splitting and merging.

(* split : int list -> int list * int list *)
(* REQUIRES true *)
(* ENSURES split(L) = a pair (A, B) of lists such that *)
(* length(A) and length(B) differ by at most 1 *)
(* and A@B is a permutation of L. *)

fun split [ ] = ([ ], [ ])
| split [x] = ([x], [ ])
| split (x::y::L) = let val (A, B) = split L in (x::A, y::B) end

Example: split [1,2,3,4,5] = ([1,3,5],[2,4])

We can prove that split meets its specification, by induction on the length of the list being split. In the proof we appeal to some obvious facts about permutations.

- For L of length ≤ 1 the result holds obviously.

- Let L be a list of length n > 1. Then we can express L in the form x::y::R for some integers x and y and a list R of shorter length than n. By induction hypothesis, split R evaluates to a pair of lists (A,B) such that length(A) and length(B) differ by at most 1, and A@B is permutation of R. But then split L = (x::A,y::B), and these two lists have the same length difference as A and B; and (x::A)@(y::B) is a permutation of x::y::(A@B), hence also a permutation of L.

- That completes the proof.

Note that the spec is a bit more precise than the informal English that we used to introduce the algorithm.

The helper function for merging is only going to be used on a pair of sorted lists, and is used to produce another sorted list containing all of the items in both of the input lists. However, it would be terribly inefficient to call the sorted function here; we don’t need to check that the inputs
are sorted lists, provided we prove that the merge function only ever gets applied to pairs of sorted lists. And we can design the merge function so that it automatically builds a sorted result, so again there is no need to verify this property explicitly by calling sorted!

(* merge : int list * int list -> int list *)
(* REQUIRES A and B are sorted lists of integers *)
(* ENSURES merge(A, B) = a sorted permutation of A@B *)

fun merge ([ ], B) = B
| merge (A, [ ]) = A
| merge (x::A, y::B) = case compare(x,y) of
    LESS => x :: merge(A, y::B)
| EQUAL => x::y::merge(A, B)
| GREATER => y :: merge(x::A, B)

Examples:

    merge([1,3,5], [2,4]) = [1,2,3,4,5]
    merge([5,3,1], [2,4]) = [2,4,5,3,1]

We prove that merge meets its spec by induction on the sum of the lengths of A and B. This strategy should work, because in each recursive call the length of at least one of the two lists decreases by 1 (and the other one is either the same as before or shorter); in all cases the sum of the two list lengths is smaller. Here are the proof details. You might expect us to use as the “base case” in this proof the case where the sum of the list lengths is 0, i.e. when both lists are empty. However, because the of the way the function is written, it is actually simpler to base our inductive case analysis on the function definition, as follows.

Note that the spec asserts that merge(A,B) is equal to a sorted perm of A@B. This is the same as asserting that merge(A,B) evaluates to a sorted perm of A@B.

* Assume that A and B are sorted lists.

* If A or B is the empty list, then merge(A,B) returns B or A, respectively. In both cases the result is sorted, and since [ ]@B = B and A@[ ] = A in each case the result is equal to (and thus a permutation of) A@B.
• Inductive step: suppose that \( A \) and \( B \) are non-empty lists, and that \( \text{merge} \) satisfies the specification on all pairs of sorted lists whose length sum is smaller than that of \( A \) and \( B \). Let \( A = x :: A' \) and \( B = y :: B' \). Just like the function definition (third clause), our proof branches on the result of comparing \( x \) and \( y \).

  - If \( x < y \), \( \text{merge}(A, B) = x :: \text{merge}(A', B) \). By assumption that \( x :: A' \) is sorted, \( x \) is \( \leq \) every item in \( A' \). And by assumption that \( A \) is sorted, so is \( A' \). The length of \( A' \) is one less than the length of \( A \). So by the induction hypothesis, \( \text{merge}(A', B) \) evaluates to a sorted list (say \( M \)) that is a permutation of \( A'@B \). We have \( \text{merge}(A, B) = x :: M \). Since \( x < y \) and we assumed that \( y :: B' \) is sorted, \( x \) is \( \leq \) every item in \( B \). So \( x \) is \( \leq \) every item in \( M \). Hence \( x :: M \) is a sorted permutation of \( x :: (A'@B) \).

  - The case analysis for when \( x = y \) or \( x > y \) is similar and we omit the details.

• We covered pairs in which one (or both) of the lists is empty in the first case; and the inductive step covers cases where both of the lists are non-empty. Thus we have shown by induction that for all sorted lists \( A, B \), \( \text{merge}(A, B) \) evaluates to a sorted permutation of \( A@B \).

**Digression**

The spec given here for \( \text{merge} \) only talks about what happens when the function gets applied to a pair of sorted lists. Of course the function can be used on any pair of integer lists whatsoever, but the specification doesn’t tell us anything about what will happen. Here’s what can happen:

```plaintext
val merge = fn : int list * int list -> int list
- merge ([2,1],[4,3]);
val it = [2,1,4,3] : int list
```

The input wasn’t a pair of sorted lists and the output wasn’t sorted. This doesn’t clash with the spec, and doesn’t cause any problems because we will only use the function by applying it to sorted lists!
Now we have the ingredients, we can define a mergesort function:

(* msort : int list -> int list *)
(* REQUIRES true *)
(* ENSURES msort(L) = a sorted permutation of L *)

fun msort [ ]  = [ ]
| msort [x] = [x]
| msort L =
  let
    val (A, B) = split L
  in
    merge(msort A, msort B)
  end

Note how closely the function definition resembles the informal algorithm description! And as promised, we've only used merge in a place where (it will be shown that) its arguments are sorted lists!

We now prove by induction on the length of \( L \) that \( msort \) meets this specification, for all integer lists \( L \). Of course, we will make use here of the results (already established) that \( \text{split} \) and \( \text{merge} \) satisfy their specifications.

Take note where (at the point labelled (*)) we establish that \( \text{merge} \) is being applied to sorted lists, which justifies our use of the proven spec for \( \text{merge} \)!

- **Base case:** When \( L \) is empty or a singleton list, \( msort \ L = L \), and this is trivially a sorted list and a permutation of \( L \).

- **Inductive step:** Assume that \( L \) is a list of length \( n > 1 \) and that \( msort \) satisfies the spec for lists of length less than \( n \). From above, we know that \( \text{split} \ L \) evaluates to a pair of lists \((A, B)\) such that \( 0 \leq \text{length}(A) - \text{length}(B) \leq 1 \) and \( A@B \) is a permutation of \( L \). Hence, the length of \( A \) and length of \( B \) are both less than length of \( L \). (The maximum possible length for \( A \) is \( n \div 2 \) if \( n \) is even, \( n \div 2 + 1 \) if \( n \) is odd, and since \( n > 1 \) in each case this is less than \( n \).) So, by the induction hypothesis, \( msort \ A \) evaluates to a sorted permutation of \( A \), and \( msort \ B \) evaluates to a sorted permutation of \( B \). So the values of \( msort \ A \) and \( msort \ B \) are sorted lists (*), and by the spec for \( \text{merge} \) it follows that \( \text{merge}(msort \ A, msort \ B) \) evaluates to a sorted permutation of \( A@B \). This must also be a permutation of \( L \), because “a permutation of a permutation is a permutation”. End of proof.
Digression

Just for interest again, consider the following slight variant:

fun msort’ [ ] = [ ]
| msort’ L =
  let
    val (A, B) = split L
  in
    merge(msort’ A, msort’ B)
  end

If we drop the singleton clause, like this, we get a function that loops on lists of length 1. Hence it also loops on any non-empty list. See where the above proof goes wrong if we try to use it to prove this code correct.

6 The Joy of Specs

The mergesort example shows the benefits of designing helper functions with clear specifications, chosen carefully to make appropriate assumptions about the arguments to which the functions will be applied, and to make strong enough assertions about the results produced by these functions.

To illustrate the potential problems caused by inappropriate helper specs, note that the merge function also satisfies the specification:

For all integer lists L and R,
  merge(L,R) evaluates to a permutation of L@R.

This spec is not strong enough to help prove that msort sorts.

Note also that we can replace split by any other function with the same type that satisfies the same specification as we used above, without affecting the correctness of msort (defined as above, but using the replacement split function). The new split function doesn’t need to be extensionally equivalent to the old one; it just needs to satisfy the same specification! For example we could have used

fun split [ ] = ([ ], [ ])
| split [x] = ([ ], [x])
| split (x::y::L) = let val (A, B) = split L in (x::A, y::B) end
Exercises

• Check that this split function satisfies the same specification as before.

• Show that this function is not extensionally equivalent to the original split. (Hint: what is split [1, 2, 3]? For which integer lists \( L \) do the two split functions produce exactly the same result?

7 Work of msort

The work (sequential running time) of \( \text{msort}(L) \) depends on the length of \( L \), but not on the integers that occur in \( L \). We can derive, from the function definition, a recurrence relation for the work \( W_{\text{msort}}(n) \) of \( \text{msort}(L) \) when \( L \) has length \( n \). To get an asymptotic estimate of the work for \( \text{msort} \), we must also analyze the work of \( \text{split} \) and \( \text{merge} \).

It seems pretty obvious that \( \text{split}(L) \) looks at each item in \( L \) successively, dealing them out into the left- or right-hand component of the pair of lists being constructed. So \( W_{\text{split}}(n) \) is \( O(n) \). We can reach the same conclusion by extracting a recurrence relation from the definition of \( \text{split} \):

\[
W_{\text{split}}(0) = c_0 \\
W_{\text{split}}(1) = c_1 \\
W_{\text{split}}(n) = c_2 + W_{\text{split}}(n - 2) \quad \text{for } n > 1
\]

for some constants \( c_0, c_1, c_2 \). It is easy to show that the solution \( W_{\text{split}}(n) \) to this recurrence relation is \( O(n) \).

Similarly, when \( A \) and \( B \) are lists of length \( m \) and \( n \), the running time of \( \text{merge}(A, B) \) is linear in \( m + n \). (The output list has length \( m + n \).)

Apart from the empty and singleton cases, \( \text{msort}(L) \) first calls \( \text{split}(L) \), then calls \( \text{msort} \) recursively twice, each time on a list of length about half of the original list’s length, then calls \( \text{merge} \) on a pair of lists whose lengths add up to \( \text{length}(L) \). Hence, the work of \( \text{msort} \) on a list of length \( n \) is given inductively by:

\[
W_{\text{msort}}(0) = k_0 \\
W_{\text{msort}}(1) = k_1 \\
W_{\text{msort}}(n) = k_2n + 2W_{\text{msort}}(n \text{ div } 2) \quad \text{for } n > 1
\]

for some constants \( k_0, k_1, k_2 \). Using the table of standard solutions, it follows that \( W_{\text{msort}}(n) \) is \( O(n \log n) \). So the work for \( \text{msort} \) on a list of length \( n \) is \( O(n \log n) \).
8 Self-test 6

1. Let $T(n)$ for $n \geq 1$ be given by the following recurrence:

\[
T(1) = 1 \\
T(n) = 2T(n \div 2) + n^2 \quad \text{for } n > 0
\]

Show that for all $k \geq 1$, $T(2^k) \leq 2^{2k+1}$. This is an easy induction! This result implies, by extrapolating, that $T(n)$ is $O(n^2)$.

Contrast what happens here with the behavior of $U(n)$, given by

\[
U(1) = 1 \\
U(n) = 2U(n \div 2) + n \quad \text{for } n > 0
\]

Show that for all $k \geq 1$, $U(2^k) \leq k2^{k+1}$.
It follows that $U(n)$ is $O(n \log n)$.

2. Prove by induction on the length of $L$ that $\text{sorted } L = \text{true}$ if every item in $L$ is $\leq$ all later items in $L$, and $\text{sorted } L = \text{false}$ otherwise. Indicate clearly when you appeal to the characteristic properties of $<$ as a linear ordering. Explain why the work to evaluate $\text{sorted } L$ when $L$ is a list value of length $n$ is $O(n)$.

3. A friend suggests that you implement a form of mergesort that checks first to see if the list is already sorted, to avoid all the recursive effort in that case. Here is their version:

```ml
fun silly_msort [ ] = [ ]
| silly_msort [x] = [x]
| silly_msort L = if (sorted L) then L else
| let
| val (A, B) = split L
| in
| merge (silly_msort A, silly_msort B)
| end
```

What’s the work for a list of length $n$?
4. Another friend suggests `new_msort`, which calls the old `msort` function if it needs to recurse:

```ml
fun new_msort [] = []
| new_msort [x] = [x]
| new_msort L = if (sorted L) then L else
  let val (A, B) = split L in merge (msort A, msort B) end
```

What's the work for `new_msort L` when `L` is a list of length `n`?

5. When the expression `ins(3,ins(2,ins(1,[ ])))` is evaluated, in what order do the insertions occur? How about `ins(1,ins(2,ins(3,[ ])))`?

6. What kind of argument values for `ins` cause the evaluation of `ins(x,L)` to take the most steps? These are called worst-case arguments for `ins`. Give a recurrence for the work of `ins(x,L)` when `L` is a list of length `n`. Solve the recurrence and give an asymptotic classification. Estimate the work for `isort L` when `L` is a list of length `n`.

7. The definition of `perms` given earlier has 4 clauses. Does it make any difference to the applicative behavior of the function if we change the clause order? Rewrite the `perms` function to use fewer than 4 clauses.

8. Write an ML function

   ```ml
   split : int list -> int list * int list
   ```

   such that for all integer lists `L`, `split(L)` evaluates to a pair of lists `(A, B)` with `A @ B = L` and `|length(A) - length(B)| ≤ 1`. Here `|x|` means the absolute value of `x`. Would the code for `msort` still satisfy the sorting specification if we replace its `split` function with this one?
9. Here is an ML implementation of the bubblesort algorithm, using recursive functions

\[
\begin{align*}
bub & : \text{int list} \rightarrow \text{int list} \\
bubsort & : \text{int list} \rightarrow \text{int list}
\end{align*}
\]

Despite its reputation as a “bad” sorting algorithm (“bubble sort seems to have nothing to recommend it”, says Knuth) we include it here because it is easy to write (elegantly) in ML and serves as a good exercise in correctness and runtime analysis.

\[
\text{local} \\
\text{ fun bub [ ]} = [ ] \\
\mid \quad \text{ fun bub [x]} = [x] \\
\mid \quad \text{ fun bub (x::y::L)} = \\
\quad \quad \quad \text{ if } y < x \text{ then } y::\text{bub(x::L)} \text{ else } x::\text{bub(y::L)} \\
\text{ end}
\]

\[
\text{ fun bubsort [ ]} = [ ] \\
\mid \quad \text{ fun bubsort (x::L)} = \text{bub (x::bubsort L)}
\]

(a) Evaluate \( \text{bubsort [3,2,1]} \).

(b) Does it make any difference to the input-output behavior of this function if we change the order of the clauses in \( \text{bub} \)? How about the asymptotic runtime? What happens if we omit the clause for \( \text{bub [x]} \)? How about dropping just the clause for \( \text{bub [ ]} \)?

(c) Say that a list is almost-sorted if it is empty, or it is non-empty and its tail is sorted. (So a sorted list is also almost-sorted!) Prove that for all integer lists \( L \), if \( L \) is almost-sorted, then \( \text{bub L} \) evaluates to a sorted permutation of \( L \). Deduce that for all integer lists \( L \), \( \text{bubsort L} \) evaluates to a sorted permutation of \( L \).

(d) Show that the work to evaluate \( \text{bubsort L} \) when \( L \) is a list of length \( n \) is \( O(n^2) \). You will need first to find the work to evaluate \( \text{bub L} \) when the length of \( L \) is \( n \). Does it make any difference for the work done by \( \text{bub} \) if you assume the list is almost-sorted?
Here is an ML implementation of selection sort, based on functions

\[
\begin{align*}
\text{sel} & : \text{int} \times \text{int list} \times \text{int list} \rightarrow \text{int} \times \text{int list} \\
\text{selsort} & : \text{int list} \rightarrow \text{int list}
\end{align*}
\]

\[
\text{local}
\begin{align*}
\text{fun sel} (x, \[\], B) & = (x, B) \\
| \text{sel} (x, a::A, B) & = \\
   \text{if} \ a < x \ \text{then} \ \text{sel} (a, A, x::B) \ \text{else} \ \text{sel} (x, A, a::B)
\end{align*}
\]
in

\[
\begin{align*}
\text{fun selsort} \ [\] & = [\] \\
| \text{selsort} (x::L) & = \\
   \text{let} \\
   \text{val} \ (y, R) & = \text{sel} (x, L, [\]) \\
   \text{in} \\
   y :: \text{selsort} R \\
   \text{end}
\end{align*}
\]

end

The idea is to keep finding the minimum element (using \text{sel}), which will be the first item in the sorted result, and sorting the rest (recursively). The \text{sel} function maintains an accumulator list (the third argument) containing all except the smallest value seen so far of the input list, the first argument is the minimum so far, and the second argument is the rest of the input list not yet examined. \text{sel} is a self-contained recursive function that serves as a helper for \text{selsort}. As there’s no reason to make this helper available elsewhere we put it into a \text{local} declaration.

Prove correctness of these functions (for suitably chosen specifications!) and estimate the work on lists of length \(n\). The main effort will be for the helper function, which (you might say, if you pardon the pun) does most of the work! \textit{We’ve left the details a bit vague here to give you a chance to explore! Choose a good spec for \text{sel} to help prove correctness of \text{selsort}. Ask if you want some hints.}